// File: contracts/OddLife.sol



// SPDX-License-Identifier: None

pragma solidity ^0.8.4;

import "@openzeppelin/contracts/access/Ownable.sol";

interface IOddContract {
  function ownerOf(uint256 tokenId) external returns (address);
  function transferFrom(address from, address to, uint256 tokenId) external;
}

contract OddLife is Ownable {
  uint256 constant PRICE = 0.01 ether;

  IOddContract private immutable oddFrensContract;
  IOddContract private immutable oddPetsContract;

  mapping(uint256 => bool) public isAlive;

  event BringToLife(uint256 indexed tokenId);

  constructor(address _oddFrensContract, address _oddPetsContract) {
    oddFrensContract = IOddContract(_oddFrensContract);
    oddPetsContract = IOddContract(_oddPetsContract);
  }

  function sendPetsToLive(uint256 _frenId, uint256[] calldata _petIds) external {
    require(
      oddFrensContract.ownerOf(_frenId) == msg.sender,
      "NOT FREN OWneR"
    );
    require(isAlive[_frenId] == false, "aLrEadY ALiVe");
    require(_petIds.length == 3, "NEEds 3 PeTs");

    for (uint256 i = 0; i < _petIds.length; ++i) {
      require(
        oddPetsContract.ownerOf(_petIds[i]) == msg.sender,
        "NOT PET OWneR"
      );

      oddPetsContract.transferFrom(msg.sender, 0x000000000000000000000000000000000000dEaD, _petIds[i]);
    }
    
    isAlive[_frenId] = true;

    emit BringToLife(_frenId);
  }

  function payToLive(uint256 _tokenId) external payable {
    require(
      oddFrensContract.ownerOf(_tokenId) == msg.sender,
      "NOT FREN OWneR"
    );
    require(isAlive[_tokenId] == false, "aLrEadY ALiVe");
    require(msg.value == PRICE, "EthEr sENt is NOT corrEcT");
    
    isAlive[_tokenId] = true;

    emit BringToLife(_tokenId);
  }

  function areAlive(uint256[] calldata _tokenIds)
    external
    view
    returns (bool[] memory)
  {
    bool[] memory alive = new bool[](_tokenIds.length);

    for (uint256 i = 0; i < _tokenIds.length; ++i) {
      alive[i] = isAlive[_tokenIds[i]];
    }

    return alive;
  }

  function devBringToLife(uint256[] calldata _tokenIds) external onlyOwner {
    for (uint256 i = 0; i < _tokenIds.length; ++i) {
      isAlive[_tokenIds[i]] = true;

      emit BringToLife(_tokenIds[i]);
    }
  }

  function withdraw() external onlyOwner {
    require(
      payable(owner()).send(address(this).balance),
      "WIThDRaW UNsucCEssFUl"
    );
  }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}